



Tips and Tricks for Sending Emails Using SAS[®]

Chris Hemedinger, SAS Communities & User Groups





Chris Hemedinger

Senior Manager, SAS Online Communities & User Groups

Chris leads the team of community managers for communities.sas.com, where SAS users gather and discuss SAS technologies, find answers to questions, share their experiences and provide feedback. He is also involved in SAS' strategy for user groups and other online communities with SAS customers. He gathers timely content for the user community, drawing more SAS experts into the conversations.

SAS: my most prolific pen pal

_SAS-generated email

From	Subject	Received
Today		
SAS Communities	SASWare Ballot Ideas: 0 recent entries as of 30AUG21 NO recent SASWare Ballot Ideas (past 5 days) <end>	Mon 8/30/2021 08:02
SAS Communities	SAS Employees in Communities: 0 added on 30AUG21 NO employees added today. <end>	Mon 8/30/2021 07:50
SAS Communities	California members: 21 ADDED on 30AUG21 The following new users were INVITED/ACCEPTED into the California Group (30AUG21)	Mon 8/30/2021 07:50
SAS Communities	anthem members: 2 ADDED on 30AUG21 The following new users were INVITED/ACCEPTED into the anthem Group (30AUG21)	Mon 8/30/2021 07:48
SAS Communities	Processed: 139996 of 139938 topics 139990 total topics, 11369 Employee topics, 121913 Public topics	Mon 8/30/2021 07:12
Chris Hemedinger	blogs.sas.com: 15 posts, 34 comments (August 25, 2021 - Augu... This automated report shows blog activity between August 25, 2021 and August 30, 2021	Mon 8/30/2021 06:20
SAS Communities	Hot fix bulletins: 4 published on 30AUG21 The following hot fix notices were published to SAS Communities on 30AUG21	Mon 8/30/2021 06:01
SAS Communities	DAILY REPORT github.com/sassoftware activity as of 30AUG2021 2-week activity on github.com/sassoftware...	Mon 8/30/2021 05:36
Last Week		
SAS Communities	SASWare Ballot Ideas: 1 recent entries as of 27AUG21 The following ideas were added to SASWare Ballot in the past 5 days (27AUG21)	Fri 8/27/2021 07:48
SAS Communities	SAS Employees in Communities: 2 added on 27AUG21 The following new user names are added to SAS Employee role (27AUG21)	Fri 8/27/2021 07:35
SAS Communities	anthem members: 8 ADDED on 27AUG21 The following new users were INVITED/ACCEPTED into the anthem Group (27AUG21)	Fri 8/27/2021 07:35
SAS Communities	Processed: 139938 of 139937 topics 139938 total topics, 11369 Employee topics, 121869 Public topics	Fri 8/27/2021 07:14
Chris Hemedinger	blogs.sas.com: 15 posts, 43 comments (August 20, 2021 - Augu... This automated report shows blog activity between August 20, 2021 and August 27, 2021	Fri 8/27/2021 06:19
SAS Communities	Hot fix bulletins: 0 published on 27AUG21 NO hot fix notices were published today. <end>	Fri 8/27/2021 06:00
SAS Communities	DAILY REPORT github.com/sassoftware activity as of 27AUG2021 2-week activity on github.com/sassoftware...	Fri 8/27/2021 05:36
SAS Communities	SASWare Ballot Ideas: 1 recent entries as of 26AUG21 The following ideas were added to SASWare Ballot in the past 5 days (26AUG21)	Thu 8/26/2021 09:01
SAS Communities	anthem members: 3 ADDED on 26AUG21 The following new users were INVITED/ACCEPTED into the anthem Group (26AUG21)	Thu 8/26/2021 07:48
SAS Communities	SAS Employees in Communities: 4 added on 26AUG21 The following new user names are added to SAS Employee role (26AUG21)	Thu 8/26/2021 07:48
SAS Communities	Processed: 139794 of 139544 topics	Thu 8/26/2021 07:12

DAILY REPORT github.com/sassoftware activity as of 30AUG2021



SAS Communities

To: Chris Hemedinger

ⓘ If there are problems with how this message is displayed, click here to view it in a web browser.

Recent pull requests

Obs	repo_link	title	state	login	created_at
1	lac-testsuite	Initial commit for logs	closed	MeganGit	17AUG2021
2	python-sascli	Gitlab merge	open	smindauer	27AUG2021
3	viva4-lac-gcp	enable and document GCR access	open	endem	27AUG2021
4	viva4-deployment	Added in support for openssl-generator	closed	thpang	24AUG2021
5	sas-visualanalytics-thirdpartyvisualizations	Fixes minor flaws in funnel and circle packing and	closed	ruppi	27AUG2021
6	restapi-demos	Bump jszip from 3.6.0 to 3.7.1	closed	dependabot[bot]	27AUG2021
7	restapi-demos	Bump url-parse from 1.5.1 to 1.5.3	closed	dependabot[bot]	27AUG2021
8	viva4-monitoring-kubernetes	Support job names and new sas_job_info metric in S	open	BryanEllington	27AUG2021
9	sas-visualanalytics-thirdpartyvisualizations	README and minor changes in export2CSV	closed	ruppi	26AUG2021
10	restapi-demos	Bump tar from 4.4.13 to 4.4.15	closed	dependabot[bot]	04AUG2021
11	sasrv	Allow SASSession to be used as a context manager	open	gim26	26AUG2021
12	deyascom-rest-api-samples	Correct example for updateState endpoint	closed	jocost	26AUG2021
13	gotagger	release: v0.6.3	closed	wfscheper	26AUG2021
14	gotagger	fix: major version revving in go modules	closed	wfscheper	26AUG2021
15	sas-visualanalytics-thirdpartyvisualizations	Fixes issue with selections	closed	ruppi	25AUG2021
16	viva4-art	(Issue 135) Pre install Check Display all Storage	closed	lasiva	12AUG2021
17	sas-visualanalytics-thirdpartyvisualizations	Adds Organizational Chart example based on Google	closed	ruppi	25AUG2021

Why use SAS to send email?



- Notify ("SAS batch job completed!")
- Send alerts ("uh oh, trouble in the data!")
- Distribute a report
- Customized, data-driven messages
- Trigger another process between systems that don't otherwise "talk"

How to send email using SAS

An email message is a great way to send a notification when a SAS job completes, or to monitor program execution when you're not around to check on the job. You can send email from SAS.

In this article:

- Send an email with SAS: a simple example
- How to configure SAS to send email to SAS
- How to send email with attachments in SAS
- How to send email with embedded images in SAS
- How to send email to multiple recipients in SAS
- How to send email to the developer SAS@SAS
- How to use SAS to check email status or send email with SAS

Sending email with SAS: a simple example

Where to learn more

"How to send email using SAS" at SAS Support Communities


FILENAME EMAIL

Write from SAS to the inbox

Code

```
/* Email message with dynamic content based on data */  
filename msg email  
  to="chuck.sasuser@company.com"  
  from="replies-disabled@company.com"  
  subject = "Created: Cars from &origin.";  
  
data _null_;  
  file msg;  
  put "%trim(&count.) cars selected, subset from  
&origin.";  
run;
```

Result



SMTP: essential for FILENAME EMAIL

simple mail transfer protocol

Email app



SAS code



SMTP



Email recipients



Internet or intranet

Setting your SMTP options

These are the most important

Check options

```
proc options group=email;  
run;
```

- EMAILSYS = SMTP
 - Not MAPI (Windows only)
- EMAILHOST = yourhost.mail.company.com
- EMAILPORT = 25 (or whatever your server uses)
 - Secure services often change default port
- EMAILID & EMAILPW (if needed)

Sending email: basic steps

Example: batch job completed

```
...  
%let msgerr = Errors in fetching users - see log.;  
...  
options emailhost='mailhost.mycompany.com' emailsys=smtp;  
filename msg email to="Chris.Hemedinger@sas.com"  
  FROM = "Communities Bot <chris.hemedinger@sas.com>"  
  subject="&totalusers. Communities User Records gathered";  
  
data _null_;  
  file msg;  
  put "Started: &sysptime., Ended: %sysfunc(datetime(),datetime20.)";  
  put "&msgerr.";  
run;
```

288215 Communities User Records gathered



Chris Hemedinger
To: Chris Hemedinger

Started: 06:30, Ended: 01SEP2021:07:28:47

```
NOTE: The file MSG is:  
      E-Mail Access Device  
Message sent  
To:      Chris.Hemedinger@sas.com  
Subject: 288215 Communities User Records gathered  
NOTE: 2 records were written to the file MSG.
```



- Simplicity: DATA _null_ and PUT statement, just like writing to a file
- Easy to include "run time" and any status/error message

FILENAME EMAIL supports most email fields

Use these to send email just like a real person

```
filename mail email
from="chris.hemedinger@sas.com"
to="chris.hemedinger@sas.com"
importance='high'
readreceipt
ct="text/html"
subject="I know what you did last summer"
sensitivity='private';
```

```
data _null_;
file mail;
put '<html><body>';
put '<h2>Uh oh</h2>';
put "<p>...and I'll keep quiet";
put "<br>for a price</p>";
put '</body></html>';
run;
```



Make email pretty with ODS

Most email clients support rich content using HTML

- Use ODS HTML or ODS tagsets.msoffice2k as a base
- Tabular output via PROC PRINT or PROC REPORT
- Use ODS TEXT statements to include additional text
- **Tip:** name your ODS target with an ID= to keep it separate from other ODS destinations

```
filename mail email ... ;
ods html(id=email) file=mail style=meadow;
/* PROC output here */
ods html(id=email) close;
```

Example: ODS with custom style and PROC REPORT

```
ods path(prepend) work.template(update);
proc template;
define style styles.myemail;
parent=styles.seaside;
class body /
    htmlstyle="border:none";
class systitleandfootercontainer /
    htmlstyle="border:none";
class page /
    htmlstyle="border:none";
end;run;

FILENAME OUTPUT EMAIL
    SUBJECT = "blogs.sas.com: &postCount posts,
        &commentCount comments"
    FROM = "Chris Hemedinger <chris.hemedinger@sas.com>"
    TO = (&toList)
    CT = 'text/html';

ods tagsets.msoffice2k(id=email)
    file=OUTPUT(title="7-day blog report")
    style=myemail;

proc report nowd data=pastweek;
title "&postCount blogs.sas.com posts";
title2 color=red "NOTE: " color=blue "View counts are sourced from Google Analytics";
columns post_date post_author blog_ID post_link ga_views comment_count ;
define post_date / group order=data 'Date posted' style(column)=[just=1];
define post_author / order 'Author' style(column)=[just=1];
...
run;
ods tagsets.msoffice2k(id=email) close;
```

blogs.sas.com: 8 posts, 17 comments (August 25, 2021 - September 01, 2021)

Chris Hemedinger
 To: Chris Hemedinger, [Redacted], [Redacted], [Redacted], [Redacted]
 Wed 9/1/2021

ⓘ If there are problems with how this message is displayed, click here to view it in a web browser.

This automated report shows blog activity between August 25, 2021 and September 01, 2021
 8 posts, 17 comments

Date posted	Author	Blog	Post	GA Views	Comments
01SEP2021	Rick Wicklin	The DO Loop	On the number of bootstrap samples		
31AUG2021	Ivan Fernando Herrera	SAS Latin America	Más analítica y más humana: así será la nueva industria financiera	217	
	Robert Allison	Graphically Speaking	Gary NC data - parks and trails	383	
30AUG2021	Ivan Fernando Herrera	SAS Latin America	Economía de datos, una apuesta de los países para impulsar el crecimiento con innovación	184	
	Rick Wicklin	The DO Loop	Bootstrap correlation coefficients in SAS	71	
	SAS Taiwan	SAS Taiwan	【程序】SAS EM 讀取檔案時，出現「應用程式無法登入伺服器...localhost...警告的Windows 觸發通知」	7	
27AUG2021	Jihye Yoo	SAS Korea Blog	SAS 시리즈 #1 포스트 판매력 시대, 플랫폼이 비즈니스를 성공적으로 이끄는 방법	268	
	Peter Styliadis	SAS Users	CAS Action! Show me the Columnids!	497	



Sending email with attachments


Use the ATTACH= option to include one or more

- Files to attach must be accessible to SAS session
- File type is implied by file extension
 - You can also specify content type (MIME type)
- For deliverability, use standard accepted types
 - CSV, xlsx, PDF, ZIP (all of these can be generated by SAS)

Example: Email with XLSX attached


```
filename out "%sysfunc(getoption(WORK))/roster.xlsx";
proc export data=sashelp.class
  file=out dbms=xlsx replace;
run;
proc sql noprint;
  select count(name)
  into: size TRIMMED from sashelp.class;
quit;
filename outbox EMAIL
  to="chris.hemedinger@sas.com"
  from="chris.hemedinger@sas.com"
  subject="Your new class list: &size members"
  attach="%sysfunc(pathname(out))";
data _null_;
  file outbox;
  put "Here's your roster!";
run;
```

As shown in Outlook Online



The screenshot shows an Excel spreadsheet with the following data:

Name	Sex	Age	Height	Weight
Alfred	M	14	69	112.5
Alice	F	13	56.5	84
Berta	F	13	65.3	98
Carol	F	14	62.8	102.5
Henry	M	14	65.5	102.5
Jane	M	12	57.3	83
Jane	F	12	59.8	84.5
Janet	F	15	62.5	112.5



Email with graphics

Instead of attaching images, include them inline

- ODS HTML5 can include image data in the HTML body
- Use PNG or JPEG
 - Scalable vector graphics (SVG) are cool, but probably won't render in email
- **Alternatively:** add images with DATA step, PUT and ATTACH (INLINED) option

```
filename myemail email
to="Jim@acme.com"
from="Wiley <wcoyote@acme.com>"
sender="Wiley <wcoyote@sas.com>"
attach=('C:\temp\sasLogo.gif'
        NAME="sasLogo" INLINED="logo"
        'C:\temp\reportToEmail.html' NAME='myreport')
subject="Embedded Image Example"
content_type="text/html";

data _null_;
file myemail;
  put '<IMG SRC="cid:logo" height="40" width="160">';
run;
```


Example: Graphics in email

```
ods graphics / imagefmt=png height=400 width=400;
```

```
filename outbox EMAIL  
to="chris.hemedinger@sas.com"  
from="chris.hemedinger@sas.com"  
subject="Gas guzzlers"  
ct="text/html";
```

```
ods html5 (id=mail) file=outbox style=pearlj  
options(bitmap_mode="inline") gtitle;  
title "CARS data: mpg vs HP";  
proc sgplot data=sashelp.cars;  
scatter x=mpg_city y=horsepower;  
run;
```

```
ods html5 (id=mail) close;
```



Sending email to multiple recipients

Use data-driven methods to build your list

```
%let cutoff = %sysfunc(today()) - 7;
/* two lists: one for Monday and one for all other days */
%let dist =
%sysfunc(ifc(%sysfunc(today(),weekday1.)=2,weekly,daily));

filename dl "/u/userid/reports/distlist_&dist..csv";
data distlist;
  infile dl dsd;
  length email $ 50;
  input email;
run;

/* create space-delimited list like: "email1" "email2" "email3" */
proc sql noprint;
select cat("'",trim(email),'")
  into :toList separated by ' ' from distlist;
quit;

FILENAME OUTPUT EMAIL
SUBJECT = "Report of the day"
FROM = "Chris Hemedinger <myaddress@sas.com>"
TO = (&toList)
CT ='text/html';
```

Message sent

To: "chris.hemedinger@sas.com"
"lisa.jones@sas.com"
"shelly.g@sas.com"
"caslee.smith@sas.com"

Sending customized emails to multiple recipients

Use SAS macro language for data-driven email messages

```
%macro sendreports(Email, Make, Model_Count);
  filename outbox EMAIL;
  data _null_;
  FILE outbox
  to("&email" )
  importance="HIGH"
  sensitivity="CONFIDENTIAL"
  subject="NHTSA: &make Recall"

  file outbox;
  put "Dear &make.,";
  put "message here";
  put "NHTSA Data Team";
run;
%mend sendreports;

data _null_;
  set car_data;
  /* will send one email message per record */
  call execute(cats('%sendreports(',Email,',',
    Make,',',
    Model_Count,')'));
run;
```

cars_data

Email	Make	Model_count
safety@honda.com	Honda	13
safety@ford.com	Ford	5
safety@toyota.com	Toyota	8

Using Gmail as the SMTP sender

This might require adjustments to your Google settings

```
options emailhost=
(
  "smtp.gmail.com"
  /* alternate: port=487 SSL */
  port=587 STARTTLS auth=plain
  /* your Gmail address */
  id="your_address@gmail.com"
  /* optional: encode PW with PROC PWENCODE */
  pw="your_password"
);

filename myemail EMAIL
  to="lucky_recipient@gmail.com"
  subject="Read SAS blogs";

data _null_;
  file myemail;
  put "Dear Friend,";
  put "I recommend that you read https://blogs.sas.com ";
  put "for lots of SAS news and tips.";
run;

filename myemail clear;
```


- "Less secure apps" setting required
- **OR** use an "App password" specific for this purpose
- 2-factor auth is best practice for your Google account
- **PROTECT** your account password or app password
- Similar approach may work for other providers

Send a text message using SAS

Using a trick: send email to an SMS service!
Search: "sms via email"

```
filename ft temp;
proc http
  url="http://yerkee.com/api/fortune"
  out=ft;
run;
libname fortune json fileref=ft;

filename sms EMAIL
  to="9198765309@txt.att.net"
  from="chris.hemedinger@sas.com";
data _null_;
  file sms;
  set fortune.root;
  put fortune;
run;
```



Q&A

Please submit your questions using the Q&A icon located in the menu at the bottom of your screen



Explore Helpful Resources

[Ask the Expert](#)

View other user webinars that provide insights into using SAS products to make your job easier.

[FREE Training](#)

Learn from home – free for 30 days. Get software labs to practice and online support if needed.

[SAS Support Communities](#)

Ask questions, get answers and share insights with SAS users.

[SAS Analytics Explorers](#)

An exclusive platform to collaborate, learn and share your expertise. Gain access to a diverse network to advance your career. Special rewards and recognition exclusively for SAS users.

[SAS Users YouTube Channel](#)

A plethora of videos on hundreds of topics, just for SAS users.

[Newsletters](#)

Get the latest SAS news plus tips, tricks and more.

[Users Groups](#)

Meet local SAS users, network and exchange ideas – virtually.

[SAS Profile](#)

If you haven't already done so, create your SAS Profile to access free training, SAS Support Communities, technical support, software downloads, newsletters and more.



Thank you
for joining us for
this SAS webinar